# 19.5.2 Green's Function for two dielectric halfspaces.

## Computation of the amplitudes in the integral representations of the four pieces.

B. Schnizer

schnizer@itp.tu-graz.ac.at

2012-03-14

## Introduction

$z = \infty$: $g_{ik} = 0.$

$\varepsilon_1$ Halfspace 1

$z = 0$: $\varepsilon_2 \, (g')_{2k} = \varepsilon_1 \, (g')_{1k}$, $(g)_{2k} = (g)_{1k}$.

$\varepsilon_2$ Halfspace 2

$z = -\infty$ : $g_{ik} = 0$

The Greens functions of the potential equation $\varepsilon_i \, \Delta \, g_{ik}(z,zp) - x^2 g_{ik} = -\delta(z - zp) \, \delta_{ik}$ is computed for a two-layer problem.
The Green's function consists of 4 pieces denoted by $g_{ik}(z, zp)$ ; the first subscript belongs to z,
the coordinate of the point of observation; the second subscript to the source point coordinate zp.
The Green's function is symmetric in z and zp.
$\varepsilon(z)$ is piecewise constant as shown in the figure above.
The Sommerfeld integral (evaluated below) is used to derive integral representations for

$$G_{ik}(\rho,\phi.z;0,0,z') = \int_0^\infty J_0(\rho\,\zeta)\, e^{-\,||z\pm z'|\zeta}\; g_{ik}(\zeta;\, z,\, z';\, \varepsilon_1, \varepsilon_2)\; d\zeta\,.$$

### ▪ The Sommerfeld integral

```
int = Integrate[BesselJ[0, ρ ξ] Exp[-c ξ], {ξ, 0, Infinity}, Assumptions → c > 0 && ρ > 0]
```

$$\frac{1}{\sqrt{c^2 + \rho^2}}$$

$$\int_0^\infty J_0(\rho\,\zeta)\, e^{-\,||z\pm z|\zeta}\, d\zeta = \frac{1}{\sqrt{(z \pm z')^2 + \rho^2}}$$

## Expressions for the 4 pieces of the Greens function.

Expressions for the "diagonal pieces" of the Greens function contain the source term and that solution of the homogeneous equation which fulfils the boundary condition at +∞ or -∞ as appropriate for the halfspace under consideration.
"Nondiagonal pieces" consist only of that solution of the homogeneous equation which fulfils the boundary at ±∞ as appropriate.

```
g11 = Exp[-ζ Abs[z - zp] ] / (ε1) + h11 Exp[- ζ z]
```

$$e^{-z\zeta}\,h11+\frac{e^{-\zeta\,\mathrm{Abs}[z-zp]}}{\varepsilon 1}$$

```
g21 = h21 Exp[ ζ z]
```

$$e^{z\zeta}\,h21$$

```
g12 = h12 Exp[ - ζ z]
```

$$e^{-z\zeta}\,h12$$

```
g22 =  Exp[-ζ Abs[z - zp] ] / ( ε2) + h22 Exp[ ζ z]
```

$$e^{z\zeta}\,h22+\frac{e^{-\zeta\,\mathrm{Abs}[z-zp]}}{\varepsilon 2}$$

## Computation of the unknown coefficients

The task of finding the unknown coefficients in the preceeding equations separates into 2 independent tasks, one for each layer containing the source.

### ■ Source in layer 2

Now the continuity conditions are formulated at the interface z = 0:

Continuity conditions at z = 0 > zp:

```
g2 = Simplify[g22 , zp < z]
g112d = (ε2 D[g2, z] /. z -> 0) == (ε1 D[g12, z] /. z -> 0)
g112n  = (g2 /. z → 0)  ==  (g12 /. z  →  0)
```

$$e^{z\zeta}\,h22+\frac{e^{(-z+zp)\zeta}}{\varepsilon 2}$$

$$\varepsilon 2\left(h22\,\zeta-\frac{e^{zp\,\zeta}\,\zeta}{\varepsilon 2}\right)==-h12\,\varepsilon 1\,\zeta$$

$$h22+\frac{e^{zp\,\zeta}}{\varepsilon 2}==h12$$

```
so2 = Solve[{gl12d, gl12n}, {h12, h22}] // Flatten // Expand // Simplify
```

$$\left\{h12 \to \frac{2\, e^{zp\, \zeta}}{\varepsilon 1 + \varepsilon 2},\ h22 \to \frac{e^{zp\, \zeta}\,(-\varepsilon 1 + \varepsilon 2)}{\varepsilon 2\,(\varepsilon 1 + \varepsilon 2)}\right\}$$

```
fg12 = g12 /. so2
```

General::spell1 : Possible spelling error: new symbol name "fg12" is similar to existing symbol "g12". More…

$$\frac{2\, e^{-z\,\zeta + zp\,\zeta}}{\varepsilon 1 + \varepsilon 2}$$

```
fg22 = g22 /. so2
```

General::spell1 : Possible spelling error: new symbol name "fg22" is similar to existing symbol "g22". More…

$$\frac{e^{-\zeta\, \mathrm{Abs}[z-zp]}}{\varepsilon 2} + \frac{e^{z\,\zeta + zp\,\zeta}\,(-\varepsilon 1 + \varepsilon 2)}{\varepsilon 2\,(\varepsilon 1 + \varepsilon 2)}$$

## Source in layer 1

Continuity conditions at z = 0 < zp :

```
(* z = 0 <  zp  *)
g1 = Simplify[g11 , zp > z]
gl21d = (ε2 D[g21, z] /. z -> 0) == (ε1 D[g1, z] /. z -> 0)
gl21n  = (g21 /. z → 0)  ==  (g1 /. z  →  0)
```

$$e^{-z\,\zeta}\, h11 + \frac{e^{(z-zp)\,\zeta}}{\varepsilon 1}$$

General::spell1 : Possible spelling error: new symbol name "gl21d" is similar to existing symbol "gl12d". More…

$$h21\, \varepsilon 2\, \zeta == \varepsilon 1 \left(-h11\, \zeta + \frac{e^{-zp\,\zeta}\, \zeta}{\varepsilon 1}\right)$$

General::spell : Possible spelling error: new symbol name "gl21n" is similar to existing symbols {gl12n, gl21d}. More…

$$h21 == h11 + \frac{e^{-zp\,\zeta}}{\varepsilon 1}$$

```
so2 = Solve[{gl21d, gl21n}, {h21, h11}] // Flatten // Expand // Simplify
```

$$\left\{h21 \to \frac{2\, e^{-zp\, \zeta}}{\varepsilon 1 + \varepsilon 2},\ h11 \to \frac{e^{-zp\, \zeta}\,(\varepsilon 1 - \varepsilon 2)}{\varepsilon 1\,(\varepsilon 1 + \varepsilon 2)}\right\}$$

```
fg11 = g11 /. so2
```

General::spell1 : Possible spelling error: new symbol name "fg11" is similar to existing symbol "g11". More…

$$\frac{e^{-\zeta\, \mathrm{Abs}[z-zp]}}{\varepsilon 1} + \frac{e^{-z\,\zeta - zp\,\zeta}\,(\varepsilon 1 - \varepsilon 2)}{\varepsilon 1\,(\varepsilon 1 + \varepsilon 2)}$$

```
fg21 = g21 /. so2
```

General::spell : Possible spelling error: new symbol name "fg21" is similar to existing symbols {fg12, g21}. More…

$$\frac{2\, e^{z\,\zeta - zp\,\zeta}}{\varepsilon 1 + \varepsilon 2}$$

## Final Result: The four pieces of the amplitudes in the integral representation of the Green's function.

- **In g11: z ⩾ 0, zp = z' ⩾ 0.**

```
Map[Factor, fg11, {3}]
```

$$\frac{e^{-\zeta \text{Abs}[z-zp]}}{\varepsilon 1} + \frac{e^{-(z+zp)\zeta}(\varepsilon 1 - \varepsilon 2)}{\varepsilon 1\ (\varepsilon 1 + \varepsilon 2)}$$

- **In g21: z ≤ 0, zp = z' ⩾ 0.**

```
Map[Factor, fg21, {2}]
```

$$\frac{2\ e^{(z-zp)\zeta}}{\varepsilon 1 + \varepsilon 2}$$

- **In g12: z ⩾ 0, zp = z' ≤ 0.**

```
Map[Factor, fg12, {2}]
```

$$\frac{2\ e^{-(z-zp)\zeta}}{\varepsilon 1 + \varepsilon 2}$$

- **In g22: z ≤ 0, zp = z' ≤ 0.**

```
Map[Factor, fg22, {3}]
```

$$\frac{e^{-\zeta \text{Abs}[z-zp]}}{\varepsilon 2} + \frac{e^{(z+zp)\zeta}(-\varepsilon 1 + \varepsilon 2)}{\varepsilon 2\ (\varepsilon 1 + \varepsilon 2)}$$